
IB013 Logické programovanie: zbierka pŕıkladov 7. mája 2012

Zadania úloh pre prednášky

1 Unifikácia

1. Pre každú z nasledujúcich unifikácíı rozhodnite, či uspeje. Ak áno, zaṕı̌ste výslednú sub-
stitúciu

(a) pondelok = pondelok

(b) ’Pondelok’ = pondelok

(c) ’pondelok’ = pondelok

(d) Pondelok = pondelok

(e) pondelok = utorok

(f) den(pondelok) = pondelok

(g) den(pondelok) = X

(h) den(X) = den(pondelok)

(i) den(pondelok,X) = den(Y,utorok)

(j) den(pondelok,X,streda) = den(Y,utorok,X)

(k) den(pondelok,X,streda) = den(Y,stvrtok)

(l) den(D) = D

(m) vikend(den(sobota),den(nedela)) = vikend(X,Y)

(n) vikend(den(sobota),X) = vikend(X,den(nedela))

2 Backtracking, rez

Fibonacciho postupnosť je postupnosť č́ısel, ktorej prvé dva členy sú rovné 1 a každý ďaľśı člen je
súčtom dvoch predchádzajúcich (fib(1) = 1, fib(2) = 1, fib(n) = fib(n− 1) + fib(n− 2), n > 2).

1. Nižšie uvedený logický program obsahuje predikát fib/2, ktorý vypoč́ıta n-tý člen Fibonac-
ciho postupnosti. Program však nie je úplne korektný. Správnu hodnotu śıce vypoč́ıta, no
ak si po jej nájdeńı vyžiadame pomocou ; ďaľsie riešenia, program sa zacykĺı (správne by
mal skončǐt s odpoveďou no). Pridajte do programu na správne miesta operátory rezu tak,
aby program v takýchto pŕıpadoch necyklil.

fib(N,1) :- N =< 2.
fib(N,X) :- N1 is N-2, fib(N1,1,1,X).
fib(0,_A,X,X).
fib(N,A,B,X) :- N1 is N-1, C is A+B, fib(N1,B,C,X).

2. Zamyslite sa a zdôvodnite, prečo nasledujúci logický program, ktorý poč́ıta n-tý člen Fi-
bonacciho postupnosti priamo poďla defińıcie, nie je vhodný (napriek tomu, že výsledok
vypoč́ıta vždy korektne)

fib(1,1).
fib(2,1).
fib(N,X) :- N>2, N1 is N-1, N2 is N-2, fib(N1,Y),fib(N2,Z),X is Y+Z.

3 Rezolúcia

1. Nájdite (ak existuje) najobecneǰśı unifikátor σ nasledujúcich literálov:

(a) p(X) a p(f(Y ))

(b) p(X, f(X)) a p(f(X), Y )

(c) p(X,Y ) a p(Z,Z)

(d) p(X, 8) a p(Y, Y )

(e) p(f(X)) a p(Z, Y )

(f) p(f(Z), g(X)) a p(Y, g(Y ))

(g) p(X, g(Z), X) a p(f(Y ), Y,W )

2. Na literáloch C1, C2 vykonajte rezolúciu a nájdite rezolventu, jednotlivé kroky (premenovanie
premenných, nájdenie najobecneǰsieho unifikátora, rezolučný prinćıp) rozṕı̌ste:

(a) C1 = {q(X),¬r(Y ), p(f(Z), f(Z))} a C2 = {n(Y ),¬r(W ),¬p(f(W ), f(W ))}
(b) C1 = {q(X),¬r(Y ), p(X,Y )} a C2 = {n(Y ),¬r(W ),¬p(f(W ), f(W ))}

3. Pomocou lineárnej rezolúcie vyvráťte

S = {{¬t(X)}, {s(1)}, {c(1), c(2),¬s(1)}, {t(0),¬c(1)}, {t(1),¬c(1)}, {t(0),¬c(2)}, {t(2),¬c(2)}}

4. Preṕı̌ste nasledujúci program v prologovskej notácii ako množinu klauzuĺı. Následne na cieli
?- a. demonštrujte lineárnu rezolúciu.

a :- b,c,d.
b :- f.
c.
d.
f.

5. Pomocou rezolučného prinćıpu vyvráťte S = {{a, b}, {c,¬b}, {¬c}, {¬a, d}, {¬e}, {e,¬d}}.
Existuje viacero možných riešeńı, skúste nájsť čo najviac z nich.

6. Uvážme nasledujúci Prologovský program a dotaz ?- a. Nakreslite zodpovedajúci SLD
strom.

a :- b,c.
a :- d.
b.
b :- d.
c.
d :- e.
d.

7. Uvážme nasledujúci Prologovský program a dotaz ?- p(X,X). Nakreslite zodpovedajúci
SLD strom.

p(X,Y) :- q(X,Z), r(Z,Y).
p(X,X) :- s(X).
q(X,b).
q(b,a).
q(X,a) :- r(a,X).
r(b,a).

s(X) :- t(X,a).
s(X) :- t(X,b).
s(X) :- t(X,X).
t(a,b).
t(b,a).

8. Do programu z predchádzajúcej úlohy pridajte 1 operátor rezu, tak aby výsledný program
uspel práve dvakrát (pomôžte si nakresleným stromom z predchádzajúcej úlohy).

Zadania úloh pre cvičenia

1 Backtracking, poč́ıtanie pomocou následńıkov

1. Naṕı̌ste predikát convert/2, ktorý prevedie svoj prvý argument (prirodzené č́ıslo) na repre-
zentáciu daného č́ısla pomocou následńıkov. Napr.:
?-convert(3,X).
X = s(s(s(0)))

2. Naṕı̌ste predikát geq/2, ktorý ako argumenty dostane 2 č́ısla zadané pomocou následńıkov
a uspeje, ak prvé č́ıslo je väčšie alebo rovné druhému. Napr.:
?-geq(s(0),s(s(0))).
no

3. Naṕı̌ste predikát sucet/3, ktorý vypoč́ıta súčet dvoch č́ısel zadaných pomocou následńıkov.
Napr.:
?- sucet(s(0),s(s(0)),X).
X = s(s(s(0)))

4. Naṕı̌ste predikát sucin/3, ktorý vypoč́ıta súčin dvoch č́ısel zadaných pomocou následńıkov.
Napr.:
?- sucin(s(s(0)),s(s(0)),X).
X = s(s(s(s(0))))

5. Zadaná je databáza faktov v tvare priamacesta(a,b), ktoré hovoria o tom, že existuje
priama (jednosmerná) cesta z miesta a do miesta b. Naṕı̌ste predikát cesta/2 tak, že
cesta(a,b) uspeje, ak existuje nejaká cesta (ktorá môže prechádzať cez ľubovǒlný počet
miest) z miesta a do miesta b. Môžete predpokladať, že cesty netvoria cykly (tzn. ak raz
od́ıdete z nejakého miesta, už sa tam nedá vrátǐt)
Rozš́ırenie 1: Upravte vaše riešenie tak, že cesta(a,b,S). uspeje, ak existuje nejaká cesta a
S je zoznam miest, cez ktoré táto cesta prechádza (ideálne v porad́ı, v akom ich prechádzame)
Rozš́ırenie 2: Upravte vaše riešenie tak, aby fungovalo aj v pŕıpade, že cesty tvoria cykly.

2 Aritmetika

1. Naṕı̌ste predikát mocnina/1, ktorý uspeje ak č́ıslo zadané ako argument je mocninou dvoj-
ky (využite fakt, že mocniny dvojky je možné opakovane bez zvyšku delǐt dvomi až kým
nedostaneme jednotku)

2. Naṕı̌ste predikát sucet/2, ktorý spoč́ıta súčet prvých N prirodzených č́ısel. Napr.:
?- sucet(5,X).
X = 15

3. Naṕı̌ste predikát nsd/3, ktorý vypoč́ıta najväčšieho spoločného delitěla dvoch č́ısel. Napr.:
?- nsd(21,49,X).
X = 7
Využite Euklidov algoritmus, ktorý je založený na pozorovańı, že najväčš́ı spoločný delitěl
č́ısel a, b kde a > b je rovnaký ako najväčš́ı spoločný delitěl č́ısel (a− b), b

4. Naṕı̌ste predikát cifsucet/2, ktorý spoč́ıta ciferný súčet zadaného č́ısla. Napr.:
?- cifsucet(12345,X).
X = 15

3 Zoznamy

1. Naṕı̌ste predikát dvojnasobok/2, ktorý uspeje ak prvky v druhom zozname sú dvojnásobkami
prvkov prvého zoznamu.
napr. dvojnasobok([5,3,2],[10,6,4]) uspeje, ale dvojnasobok([1,2,3],[2,4,5]) ne-
uspeje

2. Naṕı̌ste predikát filter/2, ktorý zo zoznamu odstráni všetky neč́ıselné hodnoty, napr.:
?- filter([5,s(0),3,a,2,A,7],X).
X = [5,3,2,7]

3. Naṕı̌ste predikát cifry/2, ktorý zo zoznamu cifier vybuduje č́ıslo, napr.:
?- cifry([2,8,0,7],X).
X = 2807

4. Naṕı̌ste predikát nty/3, ktorý vráti n-tý prvok zoznamu, napr.:
?- nty(4,[5,2,7,8,0],N).
N = 8

5. Napšte predikát rovnaju sa/2, ktorý uspeje, ak sa dva zadané zoznamy rovnajú. V pŕıpade,
že sa zoznamy nerovnajú, vyṕı̌ste dôvod, prečo je to tak (jeden zoznam je kratš́ı ako druhý,
výpis prvých 2 prvkov, ktoré sa nerovnajú...) Môžete predpokladať, že zoznamy neobsahujú
premenné. Napŕıklad:
?- rovnaju sa([5,3,7],[5,3,7]).
yes

?- rovnaju sa([5,3,7],[5,3,7,2]).
1. zoznam je kratsi
no

?- rovnaju sa([5,3,7],[5,2,3,7]).
3 sa nerovna 2
no

6. Naṕı̌ste predikát priemer/2, ktorý vypoč́ıta aritmetický priemer zadaného zoznamu. Napr.:
?- priemer([1,2,3,4,5,6],X).
X = 3.5

7. Naṕı̌ste predikát insert/3, ktorý do usporiadaného zoznamu č́ısel vlož́ı ďaľsie č́ıslo tak, aby
aj výsledný zoznam bol usporiadaný. Napr.:
?- insert(7,[1,2,3,10,12],X).
X = [1,2,3,7,10,12]

4 DC gramatiky

1. Naṕı̌ste DCG pre rozpoznávanie párnych (=sudých) binárnych č́ısel.

2. Naṕı̌ste DCG rozpoznávajúcu palindrómy pozostávajúce z ṕısmen a, b (palindróm je slo-
vo, ktoré sa č́ıta rovnako od začiatku aj od konca. Napr. [], [a,b,a], [a,b,b,a] sú
palindrómy, ale [a,b,b] palindróm nie je)

3. Naṕı̌ste DCG pre jazyk, ktorý obsahuje slová v tvare anbm kde m ≥ n (pomôcka: vyskúšajte
najprv jazyk slov v tvare anbn)

4. Naṕı̌ste DCG pre jazyk slov v tvare anb2n.

5. Naṕı̌ste DCG pre jazyk slov v tvare anbn+1cn+2.

6. Preveďte nasledujúcu gramatiku na predikáty s rozdielovými klauzulami:

s --> np, vp.
np --> det, n.
vp --> tv, np.
vp --> v.
det --> [the].
det --> [a].
det --> [every].
n --> [man].
n --> [woman].
n --> [park].
tv --> [loves].
tv --> [likes].
v --> [walks].

7. Nakreslite derivačný strom pre nejakú vetu rozpoznávanú gramatikou z predchádzajúcej
úlohy.

8. Naṕı̌ste, aké termy zodpovedajú nasledujúcim derivačným stromom:

S

Ad

Yesterday

S1

NP

John

VP

V

saw

NP

Det

a

NP

Adj

good

NP

N

movie

Exp

Sub

Exp

Add

Exp

Num

5

Exp

Mul

Exp

Num

7

Exp

Num

2

Exp

Num

4

9. Pre prvý strom z predchádzajúcej úlohy zostrojte zodpovedajúcu gramatiku s vytvárańım
derivačného stromu (tzn. tak aby derivačný strom vety “Yesterday John saw a good mo-
vie” zodpovedal znázornenému - overte si to v Prologu). Nevad́ı, ak vaša gramatika bude
generovať aj iné, nezmyselné vety.

5 Logické programovanie s obmedzujúcimi podmienkami

1. Vyriešte pomocou Prologu nasledujúci algebrogram:

KC + I = OK
+ + +
A + A = KM
----------------------------
OL + KO = LI

2. Vyriešte pomocou Prologu nasledujúcu variáciu (jednu z mnoha) známej Einsteinovej hádanky:
Je rada piatich domov, pričom každý má inú farbu. V týchto domoch žije päť ľud́ı rôznych
národnost́ı. Každý z nich chová iné zviera, rád pije iný nápoj a fajč́ı iné cigarety.

I. Brit býva v červenom dome.

II. Švéd chová psa.

III. Dán pije čaj.

IV. Zelený dom stoj́ı hneď v̌lavo od bieleho.

V. Majitěl zeleného domu pije kávu.

VI. Ten, kto fajč́ı Pall Mall, chová vtáka.

VII. Majitěl žltého domu fajč́ı Dunhill.

VIII. Človek z prostredného domu pije mlieko.

IX. Nór býva v prvom dome.

X. Ten, kto fajč́ı Blend, býva veďla toho, kto chová mačku.

XI. Ten, kto chová kone, býva veďla toho, kto fajč́ı Dunhill.

XII. Ten, kto fajč́ı Blue Master, pije pivo.

XIII. Nemec fajč́ı Prince.

XIV. Nór býva veďla modrého domu.

XV. Ten, kto fajč́ı Blend, má suseda, ktorý pije vodu.

Kto chová rybičky?

Úloha má jediné riešenie, je možné jednoznačne určǐt všetky informácie, nielen informáciu o
tom, kto chová rybičky. Najzložiteǰsou časťou úlohy teda je zvolǐt si správnu reprezentáciu
zadaných faktov, následne je zvyšok relat́ıvne jednoduchý. Ak sa vám to nedaŕı, pri riešeńı
v závere zbierky nájdete nápovedu.

Riešenia úloh pre prednášky

1 Unifikácia

1. výsledky si ľahko oveŕıte v prologu

2 Backtracking, rez

1. fib(N,1) :- N =< 2, !.
fib(N,X) :- N1 is N-2, fib(N1,1,1,X).
fib(0,_A,X,X) :- !.
fib(N,A,B,X) :- N1 is N-1, C is A+B, fib(N1,B,C,X).

2. Program je vělmi neefekt́ıvny (až exponenciálna zložitosť), pretože hodnoty predchádzajúcich
členov postupnosti sa poč́ıtajú opakovane. Najlepšie to uvid́ıte, ak si nakresĺıte strom
výpočtu pre nejaké malé N, napr. 6. Takto zaṕısaný program navyše neumožňuje opti-
malizáciu posledného volania.

3 Rezolúcia

1. (a) σ = [X/f(Y )]

(b) neexistuje

(c) σ = [X/Z, Y/Z]

(d) σ = [X/8, Y/8]

(e) neexistuje

(f) σ = [X/f(Z), Y/f(Z)]

(g) σ = [W/f(g(Z)), X/f(g(Z)), Y/g(Z)]

2. (a) Premenovanie premenných ρ = [Y/A]
C1 = {q(X),¬r(A), p(f(Z), f(Z))}, C2 = {n(Y ),¬r(W ),¬p(f(W ), f(W ))}
Najobecneǰśı unifikátor σ = [Z/W ]
C1 = {q(X),¬r(A), p(f(W ), f(W ))}, C2 = {n(Y ),¬r(W ),¬p(f(W ), f(W ))}
Z rezolučného prinćıpu dostávame C = {q(X),¬r(A), n(Y ),¬r(W )}

(b) Premenovanie premenných ρ = [Y/A]
C1 = {q(X),¬r(A), p(X,A)}, C2 = {n(Y ),¬r(W ),¬p(f(W ), f(W ))}
Najobecneǰśı unifikátor σ = [X/f(W ), A/f(W )]
C1 = {q(f(W )),¬r(f(W )), p(f(W ), f(W ))}, C2 = {n(Y ),¬r(W ),¬p(f(W ), f(W ))}
Z rezolučného prinćıpu {q(f(W )),¬r(f(W )), n(Y ),¬r(W )}

3. Jeden z možných postupov

�

{t(2)}

{c(2)}

{c(2),¬s(1)}

{¬c(1)}

{¬t(X)} {t(1),¬c(1)}

{c(1), c(2),¬s(1)}

{s(1)}

{t(2),¬c(2)}

{¬t(X)}

4. P = {P1, P2, P3, P4, P5}
P1 = {a,¬b,¬c,¬d}, P2 = {b,¬f}, P3 = {c}, P4 = {d}, P5 = {f}
C = {¬a}

�

{f}

{¬d,¬f}

{¬c,¬d,¬f}

{¬b,¬c,¬d}

{¬a} {a,¬b,¬c,¬d}

{b,¬f}

{c}

{d}

{¬f}

5. Jedno z možných riešeńı:

�

{a}

{a, c}

{a, b} {c,¬b}

{¬c}

{¬a}

{¬e} {¬a, e}

{¬a, d} {e,¬d}

6.
: −a

: −b, c

: −c

�

: −d, c

: −e, c

fail

: −c

�

: −d

: −e

fail

�

7.

: −p(X,X)

: −q(X,Z), r(Z,X)

: −r(b,X)

� [X = a]

: −r(a, b)

fail

: −r(a,X), r(a,X)

fail

: −s(X)

: −t(X, a)

� [X = b]

: −t(X, b)

� [X = a]

: −t(X,X)

fail

8. s(X) :- t(X,a). nahrad́ıme za s(X) :- t(X,a),!.

Riešenia úloh pre cvičenia

1 Backtracking, poč́ıtanie pomocou následńıkov

1. convert(0,0).
convert(X,s(Y)) :- X1 is X-1, convert(X1,Y).

2. geq(_,0).
geq(s(X),s(Y)) :- geq(X,Y).

3. sucet(0,X,X).
sucet(s(X),Y,s(Z)):-sucet(X,Y,Z).

4. sucin(_X,0,0).
sucin(X,s(Y),Z) :- sucin(X,Y,Z1), sucet(X,Z1,Z).

5. Riešenie, ktoré zahŕňa obe rozš́ırenia:

(1) :- dynamic navstivene/1.

(2) cesta(A,A,[A]).
(3) cesta(A,B,[A|S]) :-
(4) assert(navstivene(A)),
(5) priamacesta(A,X),
(6) \+ navstivene(X),
(7) cesta(X,B,S).

(8) cesta(A,_B,_S) :- retract(navstivene(A)), fail.

Pre riešenie bez druhého rozš́ırenia stač́ı vynechať riadky 1,4,6,8. Alernat́ıvne by sme mohli
druhé rozš́ırenie riešǐt pridańım ďaľsieho argumentu - zoznamu už navšt́ıvených vrcholov.
Ak by sme nechceli riešǐt ani prvé rozš́ırenie, môžeme z predikátov odstránǐt tret́ı argument.
Stoj́ı za zmienku, že úloha bez rozš́ıreńı je v podstate ekvivalentá úlohe z cvičeńı s faktami
v tvare rodic(x,y). a s predikátom potomek/2.

2 Aritmetika

1. mocnina(1).
mocnina(X) :- X mod 2 =:= 0, X1 is X // 2, mocnina(X1).

2. Funkčné riešenie:

sucet(N,S) :- sucet(N,0,S).
sucet(0,S,S) :- !.
sucet(N,X,S) :- N1 is N-1, X1 is X+N, sucet(N1,X1,S).

Lepšie riešenie (využ́ıva vzorec pre súčet aritmetickej postupnosti):

sucet(N,S) :- S is (N*(N+1)) // 2.

3. nsd(X,X,X).
nsd(A,B,X) :- A>B, A1 is A-B, nsd(A1,B,X).
nsd(A,B,X) :- B>A, B1 is B-A, nsd(A,B1,X).

Dodajme, že existuje efekt́ıvneǰsia verzia využ́ıvajúca zvyšky po deleńı.

4. cifsucet(N,CS) :- cifsucet(N,0,CS).
cifsucet(0,CS,CS) :- !.
cifsucet(N,X,CS) :- X1 is X + N mod 10, N1 is N //10, cifsucet(N1,X1,CS).

3 Zoznamy

1. dvojnasobok([],[]).
dvojnasobok([X1|T1],[X2|T2]) :- X2 is 2*X1, dvojnasobok(T1,T2).

2. filter([],[]).
filter([X|T1],[X|T2]) :- number(X), !, filter(T1,T2).
filter([_X|T1],T2) :- filter(T1,T2).

3. cifry(S,X) :- cifry(S,0,X).
cifry([],X,X).
cifry([A|T],P,X) :- P1 is 10*P + A, cifry(T,P1,X).

4. nty(N,[X|_T],Y) :- N =< 1, !, X = Y.
nty(N,[_X|T],Nty) :- N1 is N-1, nty(N1,T,Nty).

5. rovnaju_sa([],[]).
rovnaju_sa([X|T1],[X|T2]) :- !, rovnaju_sa(T1,T2).
rovnaju_sa([],_S) :- write(’1. zoznam je kratsi’), !, fail.
rovnaju_sa(_S,[]) :- write(’2. zoznam je kratsi’), !, fail.
rovnaju_sa([X|_T1],[Y|_T2]) :- write(X), write(’ sa nerovna ’), write(Y), !, fail.

Pre úplnosť dodajme, že ak by sme nepožadovali výpis dôvodu nerovnosti, vystačili by sme
si s faktom rovnaju sa(S,S). (ak S neobsahuje premenné)

6. priemer(S,X) :- priemer(S,0,0,X).
priemer([],Sum,Count,X) :- X is Sum/Count.
priemer([H|T],Sum,Count,X) :- Sum1 is Sum+H, Count1 is Count+1, priemer(T,Sum1,Count1,X).

7. insert(X,[],[X]).
insert(X,[H|T],[X,H|T]) :- X =< H, !.
insert(X,[H|T],[H|T1]) :- insert(X,T,T1).

4 DC gramatiky

1. Párne č́ısla sú práve tie, ktorých binárny zápis konč́ı nulou. Daná DCG môže vyzerať
napŕıklad nasledovne:

parne --> [0].
parne --> [1], parne.
parne --> [0], parne.

2. Výsledná gramatika môže vyzerať takto:

palindrom --> [].
palindrom --> [a].
palindrom --> [b].
palindrom --> [X],palindrom,[X], {member(X,[a,b])}.

(posledný riadok je možné nahradǐt dvoma rozṕısańım možných hodnôt X)

3. Pre jazyk anbn naṕı̌seme gramatiku

an_bn --> [].
an_bn --> [a],an_bn,[b].

Gramatiku pre jazyk anbm kde m ≥ n potom môžeme zadefinovať nasledujúcim spôsobom:

an_bm --> an_bn, same(b).

Pričom same definujeme ako

same(_X) --> [].
same(X) --> [X],same(X).

(teda trochu inak ako bol definovaný na cvičeniach)

4. an_b2n --> [].
an_b2n --> [a],an_bn,[b],[b].

5. Jedno z možných riešeńı je vělmi podobné riešeniu pre jazyk anbncn, ktoré bolo uvedené
na cvičeniach. Nižšie uvedené riešenie je založené na rovnakej myšlienke (tzn. pomocou
následńıkov), avšak je zaṕısané stručneǰśım spôsobom pomocou pomocného neterminálu
repeat:

abc --> repeat(a,X), repeat(b,s(X)), repeat(c,s(s(X))).
repeat(_,0) --> [].
repeat(A,s(X)) --> [A],repeat(A,X).

6. s(S0,S) :- np(S0,S1), vp(S1,S).
np(S0,S) :- det(S0,S1), n(S1,S).
vp(S0,S) :- tv(S0,S1), np(S1,S).
vp(S0,S) :- v(S0,S).
det([the|S],S).
det([a|S],S).
det([every|S],S).
n([man|S],S).
n([woman|S],S).
n([park|S],S).
tv([loves|S],S).
tv([likes|S],S).
v([walks|S],S).

7. záviśı od vybranej vety

8. S (Ad (Yesterday), S1 (NP ( N (John) ), VP (V (saw), NP (Det (a), NP (Adj (good), NP
(N(movie)))))))
Exp( Sub(Exp( Add(Expr(Num(5)),Exp( Mul(Expr(Num(7)),Exp( Num(2))))))),Exp( Num(4)))

9. sentence(s(Ad,S)) --> adverb(Ad), sentence1(S).
adverb(ad(yesterday)) --> [yesterday].
sentence1(s1(NP,VP)) --> noun_phrase(NP), verb_phrase(VP).
noun_phrase(np(N)) --> noun(N).
noun_phrase(np(Det,NP)) --> determiner(Det), noun_phrase(NP).
noun_phrase(np(Adj,NP)) --> adjective(Adj), noun_phrase(NP).
determiner(det(a)) --> [a].
adjective(adj(good)) --> [good].
noun(n(john)) --> [john].
noun(n(movie)) --> [movie].
verb_phrase(vp(V,NP)) --> verb(V), noun_phrase(NP).
verb(v(saw)) --> [saw].

5 Logické programovanie s obmedzujúcimi podmienkami

1. alg([K,C,I,O,A,M,L]) :-
domain([K,I,O,A,L],1,9),
domain([C,M],0,9),
all_distinct([K,C,I,O,A,M,L]),
10*K + C + I #= 10*O + K,
A + A #= 10*K + M,
10*O + L + 10*K + O #= 10*L + I,
10*K + C + A #= 10*O + L,
I + A #= 10*K + O,
10*O + K + 10*K + M #= 10*L + I,
labeling([],[K,C,I,O,A,M,L]).

2. Nápoveda: Domy si môžeme oč́ıslovať žlava doprava č́ıslami 1 až 5. Pre každú národnosť,
farbu, zviera, nápoj a cigarety si vytvoŕıme premennú, ktorá bude nadobúdať hodnoty 1 až
5 poďla pŕıslušného domu. Následne môžeme napŕıklad obmedzenie “človek z prostredného
domu pije mlieko” zaṕısať ako Mlieko #= 3 a obmedzenie “Brit býva v červenom dome”
ako Brit #= Cerveny.

Kompletné riešenie:

einstein([Nor,Brit,Sved,Dan,Nemec,
Cerveny,Zeleny,Zlty,Modry,Biely,
Caj,Kava,Mlieko,Pivo,Voda,
Pes,Vtak,Macka,Kon,Ryba,
Pallmall,Dunhill,Blend,Bluemaster,Prince]):-

Vals = [Nor,Brit,Sved,Dan,Nemec,
Cerveny,Zeleny,Zlty,Modry,Biely,
Caj,Kava,Mlieko,Pivo,Voda,
Pes,Vtak,Macka,Kon,Ryba,
Pallmall,Dunhill,Blend,Bluemaster,Prince],

domain(Vals,1,5),

all_distinct([Nor,Brit,Sved,Dan,Nemec]),
all_distinct([Cerveny,Zeleny,Zlty,Modry,Biely]),
all_distinct([Caj,Kava,Mlieko,Pivo,Voda]),
all_distinct([Pes,Vtak,Macka,Kon,Ryba]),
all_distinct([Pallmall,Dunhill,Blend,Bluemaster,Prince]),

%Brit byva v cervenom dome.
Brit #= Cerveny,
%Sved chova psa.
Sved #= Pes,
%Dan pije caj.
Dan #= Caj,
%Zeleny dom stoji hned vlavo od bieleho.
Zeleny #= Biely - 1,
%Majitel zeleneho domu pije kavu.
Zeleny #= Kava,
%Ten, kto fajci Pall Mall, chova vtaka.
Pallmall #= Vtak,
%Majitel zlteho domu fajci Dunhill.
Zlty #= Dunhill,
%clovek z prostredneho domu pije mlieko.
Mlieko #= 3,
%Nor byva v prvom dome.
Nor #= 1,
%Ten, kto fajci Blend, byva vedla toho, kto chova macku.
(Blend #= Macka+1 ; Blend #=Macka-1),
%Ten, kto chova kone, byva vedla toho, kto fajci Dunhill.
(Kon #= Dunhill+1 ; Kon#=Dunhill-1),
%Ten, kto fajci Blue Master, pije pivo.
Bluemaster #= Pivo,
%Nemec fajci Prince.
Nemec #= Prince,
%Nor byva vedla modreho domu.
(Nor #= Modry + 1 ; Nor #= Modry-1 ),
%Ten, kto fajci Blend, ma suseda, ktory pije vodu.
(Blend #= Voda + 1; Blend#= Voda-1),

labeling([],Vals).

